

SEND-EMAIL

File	212
Op-Sys	OS/390, VSE/ESA, BS2000/OSD
Statement	PROCESS
Task	This view provides the support for sending electronic mails.

Common Fields for all Operating Systems

Dictionary Field Name	F/L	Mu	DE	Remarks
ERROR-CODE	N3			
ERROR-TEXT	A58			
NODE	N5		D	
NODE-NAME	A16		D	
SYSTEM-CODE	B2			
SYSTEM-MESSAGE-CODE	A10			
FUNCTION	A8		D	
SUBJECT	A128		D	
RECORD	A253		D	
RECIPIENT	A128	M20	D	
CC-RECIPIENT	A128	M20	D	
IDENTIFIER	A8		D	Required for nested SEND-EMAIL requests.

Relevant Error Codes

Code	Text	OS/390	VSE/ESA	BS2000/OSD
530	Access denied by Security Facility.	X	X	X
699	Not enough main storage available.	X	X	
799	ENTIRE SYSTEM SERVER internal error.			X
809	Subsystem not active.	X	X	X
895	<i>RECIPIENT</i> must be specified.	X	X	X
895	<i>SUBJECT</i> must be specified.	X	X	X
896	Error occurred when executing E-Mail.			X
897	Mailer response: :1:.	X	X	X
986	No or not enough space for <i>E-mail Control Block</i> .			X
986	No or not enough space for <i>E-mail Data Block</i> .			X

Field Descriptions

Field Name	Type/Length	Operating System
FUNCTION	(A8)	OS/390, VSE/ESA, BS2000/OSD

Function to be performed. Possible options:

Option	Explanation
<blank>	Default. Write a mail record.
CLOSE	All mail records have been written. Specify this function for the last SEND-EMAIL request.

Field Name	Type/Length	Operating System
SUBJECT	(A128)	OS/390, VSE/ESA, BS2000/OSD

Specifies the subject of the mail. Field is **required** for the first SEND-EMAIL request.

Field Name	Type/Length	Operating System
RECORD	(A253)	OS/390, VSE/ESA, BS2000/OSD

Mail record to be written as line in the message body.

Field Name	Type/Length	Operating System
RECIPIENT	(A128) M20	OS/390, VSE/ESA, BS2000/OSD

Array of max. 20 recipients receiving this mail. At least one recipient is required for the first SEND-EMAIL request to send this mail.

Field Name	Type/Length	Operating System
CC-RECIPIENT	(A128) M20	OS/390, VSE/ESA, BS2000/OSD

Array of max. 20 recipients receiving a 'carbon copy' of this mail. If blank, no carbon copy will be sent. This field is optional.

Field Name	Type/Length	Operating System
IDENTIFIER	(A8)	OS/390, VSE/ESA, BS2000/OSD

Required for nested SEND-EMAIL requests. All requests for the same mail must have the same identifier.

Example

The following example illustrates the usage of the SEND-EMAIL view. The PROCESS statement is encapsulated in subroutine SUB-SEND-EMAIL.

```

DEFINE DATA LOCAL
1 SEND-EMAIL VIEW OF SEND-EMAIL
  2 ERROR-CODE
  2 ERROR-TEXT
  2 NODE
  2 NODE-NAME
  2 SYSTEM-CODE
  2 SYSTEM-MESSAGE-CODE
  2 FUNCTION
  2 SUBJECT
  2 RECORD
  2 ENCRYPT
  2 RECIPIENT (1:20)
  2 CC-RECIPIENT (1:20)
  2 IDENTIFIER
1 #FUNCTION (A008) INIT <" ">
1 #SUBJECT (A128) INIT <"Test mail">
1 #RECORD (A080)
1 REDEFINE #RECORD
  2 #RECORD1 (A040)
  2 #RECORD2 (A040)
1 #RECIPIENT (A128) INIT <"email_address"> <-- change this
1 #NODE (N005) INIT <" "> <-- change this
END-DEFINE
*
ASSIGN #RECORD1 = "Dear User,"
ASSIGN #RECORD2 = " "
PERFORM SUB-SEND-EMAIL
*
ASSIGN #RECORD1 = "this is a mail sent by a Natural test pr"
ASSIGN #RECORD2 = "ogram."
PERFORM SUB-SEND-EMAIL
*
ASSIGN #RECORD1 = " "
ASSIGN #RECORD2 = " "
PERFORM SUB-SEND-EMAIL
*
ASSIGN #RECORD1 = "Best Regards,"
ASSIGN #RECORD2 = " "
PERFORM SUB-SEND-EMAIL
*
ASSIGN #RECORD1 = "ESY Development"
ASSIGN #RECORD2 = " "
PERFORM SUB-SEND-EMAIL
*
ASSIGN #FUNCTION = 'CLOSE'
ASSIGN #RECORD1 = " "
ASSIGN #RECORD2 = " "
PERFORM SUB-SEND-EMAIL
*
DEFINE SUBROUTINE SUB-SEND-EMAIL
  PROCESS SEND-EMAIL USING NODE = #NODE
  , FUNCTION = #FUNCTION
  , SUBJECT = #SUBJECT
  , RECORD = #RECORD
  , RECIPIENT = #RECIPIENT
*
IF ERROR-CODE NE 0
  WRITE 9X '=' ERROR-CODE
  / 9X '=' ERROR-TEXT
  / 8X '=' SYSTEM-CODE
  / '=' SYSTEM-MESSAGE-CODE

```

```
NEWPAGE
STOP
END-IF
END-SUBROUTINE
END
```

Supplementary Information about SEND-EMAIL

SEND-EMAIL view implements a text-based mail client. The message body is created based on text lines specified in field RECORD. To support a varying number of text lines, SEND-EMAIL was designed as update view. The mail message body is closed by specifying FUNCTION='CLOSE'.

RECORD is defined as a 253-byte character field. The data specified in the RECORD field will be terminated by carriage return and line feed in the message body.

A zero ERROR-CODE is returned from SEND-EMAIL if the e-mail has been accepted by the SMTP server. This does not necessarily mean that the e-mail could be delivered to the specified RECIPIENT(s) and CC-RECIPIENT(s).

A character set translation is needed before sending all mail data to the configured SMTP target host. Therefore, the input EBCDIC data will be converted to ASCII code page ISO 8859-1. No special characters are supported as RECORD data.

Currently SEND-EMAIL view does not support mail attachments and data encryption. The maximum size of one e-mail is limited to 63K on BS2000/OSD.

SEND-EMAIL view uses the U.S. English Code Page (code page number 037/1) as character set for input data. The at-sign as part of fields RECIPIENT and CC-RECIPIENT is represented as hexadecimal value x'7C'. This is mapped to '@' on U.S. keyboards and to '§' on German keyboards. The combination '(a)' (left bracket, lower case a, right bracket) may be used alternatively as keyboard independent at-sign. If you encounter problems with the '@' character, use '(a)' instead. Error "ESY5897 Mailer response: Send RCPT TO failed with RC 550." indicates problems with the RECIPIENT field.

For more information about e-mail administration, see the subsection Run E-Mail Client in Common Entire System Server Features in the Entire System Server Administration Documentation. See the subsection *E-Mail Client Requirements* in the sections Installation for OS/390 (Step 15), Installation for VSE/ESA (Step 12), Installation for BS2000/OSD (Step 9) in the Entire System Server Installation Documentation for additional information about e-mail installation requirements.